

EXAMINE

This chapter covers the following topics:

- Syntax 1 - EXAMINE
- Syntax 2 - EXAMINE TRANSLATE
- Syntax 3 - EXAMINE for Unicode Graphemes
- Examples

Related Statements: ADD | COMPRESS | COMPUTE | DIVIDE | MOVE | MOVE ALL | MULTIPLY | RESET | SEPARATE | SUBTRACT

Belongs to Function Group: *Arithmetic and Data Movement Operations*

Syntax 1 - EXAMINE

<pre> EXAMINE [FULL [VALUE [OF]]] { operand1 { <u>SUBSTRING</u> (operand1,operand2,operand3) } [FOR] [FULL [VALUE [OF]]] [PATTERN] operand4 [DELIMITERS-option] {[DELETE-REPLACE-clause] [GIVING-clause]} </pre>

For an explanation of the symbols used in the syntax diagram, see *Syntax Symbols*.

Syntax Description - Syntax 1

The EXAMINE statement is used to observe the content of an alphanumeric or binary field, or a range of fields within an array, and to

- return the number of how many times a search-pattern was found;
- return the byte position where a search-pattern appears first;
- return the significant content length of a field; that is, the field length without trailing blanks;
- return the occurrence number (indices) of an array field, where a pattern was found first;
- replace a pattern by another pattern;
- delete a pattern.

Operand Definition Table:

Operand	Possible Structure				Possible Formats										Referencing Permitted	Dynamic Definition			
<i>operand1</i>	C*	S	A		A	U							B					yes	no
<i>operand2</i>	C	S						N	P	I			B*					yes	no
<i>operand3</i>	C	S						N	P	I			B*					yes	no
<i>operand4</i>	C	S			A	U							B					yes	no

* *operand1* can only be a constant if the GIVING clause is used, but not if the DELETE/REPLACE clause is used.

* Format B of *operand2* and *operand3* may be used only with a length of less than or equal to 4.

Syntax Element Description:

<i>operand1</i>	<p><i>operand1</i> is the field whose content is to be examined.</p> <p>If <i>operand1</i> is a DYNAMIC variable, a REPLACE operation may cause its length to be increased or decreased; a DELETE operation may cause its length to be set to zero. The current length of a DYNAMIC variable can be ascertained by using the system variable *LENGTH.</p>
<i>operand4</i>	<p><i>operand4</i> is the value to be used for the examine operation.</p>
FULL	<p>If FULL is specified for an operand, the entire value, including trailing blanks, will be processed. If FULL is not specified, trailing blanks in the operand will be ignored.</p>
SUBSTRING	<p>Normally, the content of a field is examined from the beginning of the field to the end or to the last non-blank character.</p> <p>With the SUBSTRING option, you examine only a certain part of the field. After the field name (<i>operand1</i>) in the SUBSTRING clause, you specify first the starting position (<i>operand2</i>) and then the length (<i>operand3</i>) of the field portion to be examined.</p> <p>For example, to examine the 5th to 12th position inclusive of a field #A, you would specify:</p> <pre>EXAMINE SUBSTRING(#A, 5, 8) .</pre> <p>Notes:</p> <ol style="list-style-type: none"> 1. If you omit <i>operand2</i>, the starting position is assumed to be 1. 2. If you omit <i>operand3</i>, the length is assumed to be from the starting position to the end of the field. 3. If SUBSTRING is used in conjunction with a DYNAMIC variable, the field behaves like a fixed length variable; that is, the length (*LENGTH) does not change as a result of the EXAMINE operation, regardless of whether a DELETE or REPLACE operation was executed or not.

PATTERN	<p>If you wish to examine the field for a value which contains "wild characters", that is symbols for positions not to be examined, you use the PATTERN option. <i>operand4</i> may then include the following symbols for positions to be ignored:</p> <ul style="list-style-type: none"> ● A period (.), question mark (?) or underscore (_) indicates a single position that is not to be examined. ● An asterisk (*) or a percent sign (%) indicates any number of positions not to be examined. <p>Example: With PATTERN 'NAT*AL' you could examine the field for any value which contains NAT and AL no matter which and how many other characters are between NAT and AL (this would include the values NATURAL and NATIONAL as well as NATAL).</p>
DELIMITERS-option	This option is used to scan for a value which exhibits delimiters. For details, see <i>DELIMITERS Option</i> below.
DELETE-REPLACE-clause	The DELETE option of this clause is used to delete each search-value (<i>operand4</i>) found in <i>operand1</i> , whereas the REPLACE option is used to replace each search-value (<i>operand4</i>) found in <i>operand1</i> by the value specified in <i>operand6</i> . For details, see <i>DELETE REPLACE Clause</i> below.
GIVING-clause	For details, see <i>GIVING Clause</i> below.

DELIMITERS Option

{ ABSOLUTE [WITH DELIMITERS] [WITH DELIMITERS] <i>operand5</i> }
--


Operand Definition Table:

Operand	Possible Structure			Possible Formats								Referencing Permitted	Dynamic Definition
<i>operand5</i>	C	S		A			B					yes	no

Syntax Element Description:

ABSOLUTE	This is the default option. It results in an absolute scan of the field for the specified value regardless of what other characters may surround the value.
WITH DELIMITERS	Is used to scan for a value which is delimited by blanks or by any characters that are neither letters nor numeric characters.
WITH DELIMITERS <i>operand5</i>	Is used to scan for a value which is delimited by the character(s) specified in <i>operand5</i> .

DELETE REPLACE Clause



Operand Definition Table:

Operand	Possible Structure				Possible Formats										Referencing Permitted	Dynamic Definition
<i>operand6</i>	C	S			A	U			B						yes	no


Syntax Element Description:

DELETE	Is used to delete the first (or all) occurrence(s) of the search-value (<i>operand4</i>) in the content of <i>operand1</i> .
REPLACE	Is used to replace the first (or all) occurrence(s) of the search-value (<i>operand4</i>) in <i>operand1</i> by the replace value specified in <i>operand6</i> .
FIRST	If you specify the keyword FIRST , only the first identical value will be deleted/replaced.

Notes:

1. If the REPLACE operation results in more characters being generated than will fit into *operand1*, you will receive an error message.
2. If *operand1* is a dynamic variable, a REPLACE operation may cause its length to be increased or decreased; a DELETE operation may cause its length to be set to zero. The current length of a dynamic variable can be ascertained by using the system variable *LENGTH. For general information on dynamic variables, see *Using Dynamic Variables*.

GIVING Clause



Operand Definition Table:

Operand	Possible Structure	Possible Formats	Referencing Permitted	Dynamic Definition
<i>operand7</i>	S	N P I	yes	yes

Syntax Element Description:

GIVING	If only the keyword GIVING is specified, this corresponds to GIVING NUMBER (default).
NUMBER	Is used to obtain information on how many times the search value (<i>operand4</i>) was found in the field (<i>operand1</i>) whose content is to be examined.
POSITION	Is used to obtain the byte position within <i>operand1</i> (or the substring of <i>operand1</i>) where the first value identical to <i>operand4</i> was found.
LENGTH	Is used to obtain the remaining content length of <i>operand1</i> (or the substring of <i>operand1</i>) after all delete/replace operations have been performed. Trailing blanks are ignored.
<i>operand7</i>	The number of occurrences of the search-value. If the REPLACE FIRST or DELETE FIRST option is also used, the number will not exceed 1.
INDEX <i>operand7...3</i>	See below.

GIVING INDEX

[GIVING] INDEX [IN] <i>operand7 ... 3</i>

This option is only applicable if the underlying field to be examined is an array field.


Syntax Element Description:

INDEX	GIVING INDEX is used to obtain the array occurrence number (index) of <i>operand1</i> in which the first search-value (<i>operand4</i>) was found.
<i>operand7...3</i>	<i>operand7</i> must be specified as many times as there are dimensions in <i>operand1</i> (maximum three times). <i>operand7</i> will return 0 if the search-value is found in none of the occurrences.

Note:

If the index range of *operand1* includes the occurrence 0 (e.g. 0:5), a value of 0 in *operand7* is ambiguous. In this case, an additional GIVING NUMBER clause should be used to ascertain whether the search-value was actually found or not.

Syntax 2 - EXAMINE TRANSLATE



For an explanation of the symbols used in the syntax diagram, see *Syntax Symbols*.

Syntax Description - Syntax 2

The EXAMINE TRANSLATE statement is used to translate the characters contained in a field into upper-case or lower-case, or into other characters.

Operand Definition Table:

Operand	Possible Structure			Possible Formats										Referencing Permitted	Dynamic Definition	
<i>operand1</i>	S	A		A				B							yes	no
<i>operand2</i>	C	S			N	P	I	B*							yes	no
<i>operand3</i>	C	S			N	P	I	B*							yes	no
<i>operand4</i>	S	A		A				B							yes	no

*Format B of *operand2* and *operand3* may be used only with a length of less than or equal to 4.

Syntax Element Description:

<p>SUBSTRING <i>operand1 operand2 operand3</i></p>	<p>Normally, the content of a field is examined from the beginning of the field to the end or to the last non-blank character.</p> <p>With the SUBSTRING option, you examine only a certain part of the field. After the field name (<i>operand1</i>) in the SUBSTRING clause, you specify first the starting position (<i>operand2</i>) and then the length (<i>operand3</i>) of the field portion to be examined. <i>operand2</i> and <i>operand3</i> are specified in terms of code units.</p> <p>For example, to examine the 5th to 12th position inclusive of a field #A, you would specify:</p> <pre>EXAMINE SUBSTRING (#A, 5, 8)</pre> <p>Notes:</p> <ol style="list-style-type: none"> 1. If you omit <i>operand2</i>, the starting position is assumed to be 1. 2. If you omit <i>operand3</i>, the length is assumed to be from the starting position to the end of the field. 3. If SUBSTRING is used in conjunction with a DYNAMIC variable, the field behaves like a fixed length variable; that is, the length (*LENGTH) does not change as a result of the EXAMINE operation, regardless of whether a DELETE or REPLACE operation was executed or not.
<p>CHARPOSITION <i>operand4</i></p>	<p><i>operand4</i> defines the starting position (in terms of Unicode graphemes) of the grapheme sequence. The according position in terms of code units is returned in <i>operand6</i>. This clause can be omitted if the CHARLENGTH clause is specified; in this case the starting position 1 is assumed.</p>
<p>CHARLENGTH <i>operand5</i></p>	<p><i>operand5</i> defines the length (in terms of Unicode graphemes) of the grapheme sequence. The length of the grapheme sequence in terms of code units is returned in <i>operand7</i>. This clause can be omitted if the CHARPOSITION clause is specified; in this case the length from the starting position up to the end of the string is returned.</p>
<p>GIVING POSITION IN <i>operand6</i></p>	<p><i>operand6</i> receives the starting position (in terms of code units) of the grapheme sequence defined by <i>operand4</i> and <i>operand5</i>. If <i>operand1</i> has less than <i>operand4</i> graphemes, 0 is returned. This clause can be omitted if the GIVING LENGTH clause is specified.</p>
<p>GIVING LENGTH IN <i>operand7</i></p>	<p><i>operand7</i> receives the length (in terms of code units) of the grapheme sequence defined by <i>operand4</i> and <i>operand5</i>. If <i>operand1</i> has less than <i>operand4+operand5</i> graphemes, 0 is returned. This clause can be omitted if the GIVING POSITION clause is specified.</p>

Notes:

1. Either the CHARPOSITION or the CHARLENGTH clause or both must be specified.
2. Either the GIVING POSITION or GIVING LENGTH clause or both must be specified.

Examples

- Example 1 - EXAMINE
- Example 2 - EXAMINE SUBSTRING, PATTERN, TRANSLATE
- Example 3 - EXAMINE TRANSLATE
- Example 4 - EXAMINE for Unicode Graphemes

Example 1 - EXAMINE

```

** Example 'EXMEX1': EXAMINE
*****
DEFINE DATA LOCAL
1 #TEXT (A40)
1 #A (A1)
1 #START (N2)
1 #NMB1 (N2)
1 #NMB2 (N2)
1 #NMB3 (N2)
1 #NMBEX2 (N2)
1 #NMBEX3 (N2)
1 #NMBEX4 (N2)
1 #POSEX5 (N2)
1 #LGHEX6 (N2)
1 #NMBEX7 (N2)
1 #NMBEX8 (N2)
END-DEFINE
*
WRITE 'EXAMPLE 1 (GIVING NUMBER, WITH DELIMITER)'
MOVE 'ABC A B C .A. .B. .C. -A- -B- -C- ' TO #TEXT
ASSIGN #A = 'A'
EXAMINE #TEXT FOR #A GIVING NUMBER #NMB1
EXAMINE #TEXT FOR #A WITH DELIMITER GIVING NUMBER #NMB2
EXAMINE #TEXT FOR #A WITH DELIMITER '.' GIVING NUMBER #NMB3
WRITE NOTITLE '=' #NMB1 '=' #NMB2 '=' #NMB3
*
WRITE / 'EXAMPLE 2 (WITH DELIMITER, REPLACE, GIVING NUMBER)'
WRITE '=' #TEXT
EXAMINE #TEXT FOR #A WITH DELIMITER '-' REPLACE WITH '*'
GIVING NUMBER #NMBEX2
WRITE '=' #TEXT '=' #NMBEX2
*
WRITE / 'EXAMPLE 3 (REPLACE, GIVING NUMBER)'
WRITE '=' #TEXT
EXAMINE #TEXT ' ' REPLACE WITH '+' GIVING NUMBER #NMBEX3
WRITE '=' #TEXT '=' #NMBEX3
*
WRITE / 'EXAMPLE 4 (FULL, REPLACE, GIVING NUMBER)'
WRITE '=' #TEXT
EXAMINE FULL #TEXT ' ' REPLACE WITH '+' GIVING NUMBER #NMBEX4
WRITE '=' #TEXT '=' #NMBEX4
*

```

```

WRITE / 'EXAMPLE 5 (DELETE, GIVING POSITION)'
WRITE '=' #TEXT
EXAMINE #TEXT '+' DELETE GIVING POSITION #POSEX5
WRITE '=' #TEXT '=' #POSEX5
*
WRITE / 'EXAMPLE 6 (DELETE, GIVING LENGTH)'
WRITE '=' #TEXT
EXAMINE #TEXT FOR 'A' DELETE GIVING LENGTH #LGHEX6
WRITE '=' #TEXT '=' #LGHEX6
*
*
NEWPAGE
*
MOVE 'ABC  A B C  .A.  .B.  .C.  -A-  -B-  -C- ' TO #TEXT
*
ASSIGN #A = 'A B C'
ASSIGN #START = 6
*
WRITE / 'EXAMPLE 7 (SUBSTRING, GIVING NUMBER)'
WRITE '=' #TEXT
EXAMINE SUBSTRING(#TEXT,#START,9) FOR #A GIVING NUMBER #NMBEX7
WRITE '=' #TEXT '=' #NMBEX7
*
WRITE / 'EXAMPLE 8 (PATTERN, GIVING NUMBER)'
WRITE '=' #TEXT
EXAMINE #TEXT FOR PATTERN '-A-' GIVING NUMBER #NMBEX8
WRITE '=' #TEXT '=' #NMBEX8
*
END

```

Output of Program EXMEX1:

```

EXAMPLE 1 (GIVING NUMBER, WITH DELIMITER)
#NMB1:  4 #NMB2:  3 #NMB3:  1

EXAMPLE 2 (WITH DELIMITER, REPLACE, GIVING NUMBER)
#TEXT: ABC  A B C  .A.  .B.  .C.  -A-  -B-
#TEXT: ABC  A B C  .A.  .B.  .C.  -*  -B- #NMBEX2:  1

EXAMPLE 3 (REPLACE, GIVING NUMBER)
#TEXT: ABC  A B C  .A.  .B.  .C.  -*  -B-
#TEXT: ABC+++A+B+C+++A.++.B.++.C.++++-*---B- #NMBEX3:  18

EXAMPLE 4 (FULL, REPLACE, GIVING NUMBER)
#TEXT: ABC+++A+B+C+++A.++.B.++.C.++++-*---B-
#TEXT: ABC+++A+B+C+++A.++.B.++.C.++++-*---B-+ #NMBEX4:  1

EXAMPLE 5 (DELETE, GIVING POSITION)
#TEXT: ABC+++A+B+C+++A.++.B.++.C.++++-*---B-+
#TEXT: ABCABC.A..B..C.-*--B- #POSEX5:  4

EXAMPLE 6 (DELETE, GIVING LENGTH)
#TEXT: ABCABC.A..B..C.-*--B-
#TEXT: BCBC...B..C.-*--B- #LGHEX6:  18

EXAMPLE 7 (SUBSTRING, GIVING NUMBER)
#TEXT: ABC  A B C  .A.  .B.  .C.  -A-  -B-
#TEXT: ABC  A B C  .A.  .B.  .C.  -A-  -B- #NMBEX7:  1

EXAMPLE 8 (PATTERN, GIVING NUMBER)
#TEXT: ABC  A B C  .A.  .B.  .C.  -A-  -B-
#TEXT: ABC  A B C  .A.  .B.  .C.  -A-  -B- #NMBEX8:  1

```

Example 2 - EXAMINE SUBSTRING, PATTERN, TRANSLATE

```

** Example 'EXMEX2': EXAMINE TRANSLATE
*****
DEFINE DATA LOCAL
1 #TEXT (A50)
1 #TAB (A2/1:10)
1 #START (N2)
END-DEFINE
*
MOVE 'ABC  A B C  .A.  .B.  .C.  -A-  -B-  -C- ' TO #TEXT
*
MOVE 'AX' TO #TAB(1)
MOVE 'BY' TO #TAB(2)
MOVE 'CZ' TO #TAB(3)
*
*
WRITE 'EXAMPLE 1 (USING TRANSLATION TABLE)'
WRITE '=' #TEXT
EXAMINE #TEXT TRANSLATE USING #TAB(*)
WRITE NOTITLE '=' #TEXT
*
WRITE / 'EXAMPLE 2 (USING INVERTED TRANSLATION TABLE)'
WRITE '=' #TEXT
EXAMINE #TEXT TRANSLATE USING INVERTED #TAB(*)
WRITE NOTITLE '=' #TEXT
*
WRITE / 'EXAMPLE 3 (USING SUBSTRING, LOWER CASE)'
WRITE '=' #TEXT
ASSIGN #START = 13
EXAMINE SUBSTRING(#TEXT,#START,15) TRANSLATE INTO LOWER CASE
WRITE '=' #TEXT
END

```

Output of Program EXMEX2:

```

EXAMPLE 1 (USING TRANSLATION TABLE)
#TEXT: ABC  A B C  .A.  .B.  .C.  -A-  -B-  -C-
#TEXT: XYZ  X Y Z  .X.  .Y.  .Z.  -X-  -Y-  -Z-

EXAMPLE 2 (USING INVERTED TRANSLATION TABLE)
#TEXT: XYZ  X Y Z  .X.  .Y.  .Z.  -X-  -Y-  -Z-
#TEXT: ABC  A B C  .A.  .B.  .C.  -A-  -B-  -C-

EXAMPLE 3 (USING SUBSTRING, LOWER CASE)
#TEXT: ABC  A B C  .A.  .B.  .C.  -A-  -B-  -C-
#TEXT: ABC  A B C  .a.  .b.  .c.  -A-  -B-  -C-

```

Example 3 - EXAMINE TRANSLATE

```

** Example 'EXMEX2': EXAMINE TRANSLATE
*****
DEFINE DATA LOCAL
1 #TEXT (A50)
1 #TAB (A2/1:10)
1 #START (N2)
END-DEFINE
*
MOVE 'ABC  A B C  .A.  .B.  .C.  -A-  -B-  -C- ' TO #TEXT
*
MOVE 'AX' TO #TAB(1)

```

```

MOVE 'BY' TO #TAB(2)
MOVE 'CZ' TO #TAB(3)
*
*
WRITE 'EXAMPLE 1 (USING TRANSLATION TABLE)'
WRITE '=' #TEXT
EXAMINE #TEXT TRANSLATE USING #TAB(*)
WRITE NOTITLE '=' #TEXT
*
WRITE / 'EXAMPLE 2 (USING INVERTED TRANSLATION TABLE)'
WRITE '=' #TEXT
EXAMINE #TEXT TRANSLATE USING INVERTED #TAB(*)
WRITE NOTITLE '=' #TEXT
*
WRITE / 'EXAMPLE 3 (USING SUBSTRING, LOWER CASE)'
WRITE '=' #TEXT
ASSIGN #START = 13
EXAMINE SUBSTRING(#TEXT,#START,15) TRANSLATE INTO LOWER CASE
WRITE '=' #TEXT
END

```

Output of Program EXMEX2:

```

EXAMPLE 1 (USING TRANSLATION TABLE)
#TEXT: ABC   A B C   .A. .B. .C.   -A- -B- -C-
#TEXT: XYZ   X Y Z   .X. .Y. .Z.   -X- -Y- -Z-

EXAMPLE 2 (USING INVERTED TRANSLATION TABLE)
#TEXT: XYZ   X Y Z   .X. .Y. .Z.   -X- -Y- -Z-
#TEXT: ABC   A B C   .A. .B. .C.   -A- -B- -C-

EXAMPLE 3 (USING SUBSTRING, LOWER CASE)
#TEXT: ABC   A B C   .A. .B. .C.   -A- -B- -C-
#TEXT: ABC   A B C   .a. .b. .c.   -A- -B- -C-

```

Example 4 - EXAMINE for Unicode Graphemes

This example demonstrates the analysis of a Unicode string containing the characters ä und ü. Both characters are defined as base character followed by a combining character: ä is coded with U+0061 followed by U+0308, and ü is coded with U+0075 followed by U+0308.

```

DEFINE DATA LOCAL
1 #U (U20)
1 #START (I2)
1 #POS (I2)
1 #LEN (I2)
END-DEFINE
#U := U'AB'-UH'00610308'-U'CD'-UH'00750308'-U'EF'
*
REPEAT
  #START := #START + 1
  EXAMINE #U FOR CHARPOSITION #START
           CHARLENGTH 1
           GIVING POSITION IN #POS
           LENGTH IN #LEN
*
  INPUT (AD=0) MARK POSITION #POS IN FIELD *#U
  '          UNICODE-STRING:' #U (AD=MI)
// '          CHARACTER NO.:' #START (EM=9)
/ 'STARTS AT BYTE POSITION:' #POS (EM=9)

```

```

/ '      AND THE LENGTH IS:' #LEN      (EM=9)
WHILE #POS NE 0
END-REPEAT
END
    
```

Output:

Mainframe Environments:	Windows, UNIX and OpenVMS Environments (with Natural Web I/O Interface):
<pre> UNICODE-STRING: ABa?CDu?EF CHARACTER NO.: 1 STARTS AT BYTE POSITION: 1 AND THE LENGTH IS: 1 </pre>	<pre> UNICODE-STRING: ABäCDüEF CHARACTER NO.: 1 STARTS AT BYTE POSITION: 1 AND THE LENGTH IS: 1 </pre>
<p>Press ENTER to continue.</p>	
<pre> UNICODE-STRING: ABa?CDu?EF CHARACTER NO.: 2 STARTS AT BYTE POSITION: 2 AND THE LENGTH IS: 1 </pre>	<pre> UNICODE-STRING: ABäCDüEF CHARACTER NO.: 2 STARTS AT BYTE POSITION: 2 AND THE LENGTH IS: 1 </pre>
<p>Press ENTER to continue.</p>	
<p>Note that the character in position 3 is a combining character sequence and is two code units long.</p>	
<pre> UNICODE-STRING: ABa?CDu?E CHARACTER NO.: 3 STARTS AT BYTE POSITION: 3 AND THE LENGTH IS: 2 </pre>	<pre> UNICODE-STRING: ABäCDüEF CHARACTER NO.: 3 STARTS AT BYTE POSITION: 3 AND THE LENGTH IS: 2 </pre>
<p>And so on.</p>	